CHAPTER 6

# Labels on Maps

**Purpose:** This chapter demonstrates a variety of ways to annotate labels on maps.

---

## Labels at Hard-Coded Positions

Unfortunately there is no built-in labeling feature in Proc Gmap, and therefore if you want to add labels to a map you must do some extra programming to annotate them.

If you're just wanting to do a "quick and dirty" one-off map, hard-coding the X/Y positions for the annotated text might the easiest way to go. You just estimate the X and Y coordinates (from 0 to 100), annotate the text at those locations, look at the results, perhaps change the X and Y slightly via trial-and-error … and then call it a day.

For example, let's say we want to create a U.S. map, with the West/Central/East portions of the map labeled. These areas are somewhat 'vague' and there is no specific state or latitude/longitude that the labels need to be positioned over. So let's position some annotated labels at y=50 (50% up from the bottom of the map), and at x=20, 50, and 80 (% positions from the left of the map).

```
data anno_text;
length text $50;
xsys='3'; ysys='3'; hsys='3'; when='a';
function='label'; position='5'; style='albany amt/bold';
color='blue'; size=5;
y=50;
x=20; text='West'; output;
x=50; text='Central'; output;
x=80; text='East'; output;
run;
```

At first glance, some of the code above might look a bit obscure, but it's really pretty simple and intuitive. The xsys and ysys tell annotate what coordinate system to use – in this case the value is set to '3' so we can specify X and Y as % values. Hsys='3' lets us

specify the size of the text as a percentage of the page. And the position='5' centers the text on the specified X/Y coordinate.

anno_text

| text | xsys | ysys | hsys | when | function | position | style | color | size | y | x |
|---|---|---|---|---|---|---|---|---|---|---|---|
| West | 3 | 3 | 3 | a | label | 5 | albany amt/bold | blue | 5 | 50 | 20 |
| Central | 3 | 3 | 3 | a | label | 5 | albany amt/bold | blue | 5 | 50 | 50 |
| East | 3 | 3 | 3 | a | label | 5 | albany amt/bold | blue | 5 | 50 | 80 |

Now that we have an annotate dataset with the commands to place the text at the desired locations, it's a simple matter to specify it using the anno= option.

```
title1 ls=1.5 "Map with Annotated Text";
pattern1 c=cxFFE1FF v=s;
proc gmap map=maps.us data=maps.us anno=anno_text;
id state;
choro state / levels=1 nolegend coutline=graycc;
run;
```




Text Labels at Page Coordinates

I mentioned before that this technique works OK for a one-off map … but there are several gotchas if you decide to modify the map sometime later. If you make any changes that move the map on the page (such as adding titles or footnotes), then the labels are not going to appear in the same position – for example, if you add more titles then the map is going to be pushed down farther on the page, and the annotated text will therefore appear farther 'north' on the map. Also if you subset the map, the resulting subset would of course be at different locations on the page … but your annotated labels would still be in their original hard-coded locations.

Therefore it is preferable to place the labels on the map in a more data-driven way, so that if the map moves around, the annotated text moves with it. Which leads us to the next technique.

## Labels at Calculated Centroids

A 'map polygon' is simply an area in a map – in a world map the polygons might represent countries, a US map might have polygons for each state, and a state map might have polygons for each county. The polygons are described by X/Y coordinates all along the border of the polygons. We have a tool in SAS to calculate the approximate X/Y centroid of each polygon, and you can then annotate your text at those centroids.

Let's use the county map of Maryland to demonstrate this technique. The following code subsets Maryland out of the US county map, and then runs the %centroid macro to calculate the X/Y centers of each county, and store the results in the county_centroids dataset. The centroid macro doesn't calculate the 'perfect' visual centroid (that would be very difficult), but it is generally better than using the simple average X/Y of the polygon.

```
data my_map;
 set mapsgfk.us_counties (where=(statecode='MD'));
run;
%annomac;
%centroid(my_map,county_centroids,statecode county);
```

Here's what the first few observations of the resulting dataset looks like. Note that there is one row of data for each county.

county_centroids

| STATECODE | COUNTY | x | y |
|---|---|---|---|
| MD | 1 | 26042.92 | 759.695 |
| MD | 3 | 26110.01 | 736.386 |
| MD | 5 | 26110.56 | 754.987 |
| MD | 9 | 26112.14 | 719.799 |
| MD | 11 | 26138.24 | 733.096 |

Now let's use those X/Y coordinates, and create an annotate dataset (very similar to the one in the previous example) to annotate some text at those centroids. First, let's just annotate the '*' character at each centroid, so you can see exactly where those centroids are located.

```
data anno_labels; set county_centroids;
xsys='2'; ysys='2'; hsys='3'; when='a';
```

```
function='label'; position='5'; size=3.0; color='blue';
text='*';
run;

title1 ls=1.5 "Labels at Calculated Centroids";
pattern1 c=cxffffaa v=s;
proc gmap data=my_map map=my_map anno=anno_labels density=2;
id state county;
choro state / levels=1 nolegend coutline=graycc;
run;
```




Simple 1-Character Text

Now, instead of labeling the counties with a '*', let's merge in the county names (stored in the idname variable) from the mapsgfk.us_counties_attr dataset, and use that as the text label.

```
proc sql;
create table county_centroids as
select county_centroids.*, us_counties_attr.idname
from county_centroids left join mapsgfk.us_counties_attr
on (county_centroids.statecode=us_counties_attr.statecode)
  and (county_centroids.county=us_counties_attr.county);
```

```
quit; run;

data anno_labels; set county_centroids;
xsys='2'; ysys='2'; hsys='3'; when='a';
function='label'; position='5'; size=3.0; color='blue';
text=trim(left(idname));
run;
```




Now the map (above) has the desired text centered at the calculated centroid, but it's still not quite perfect. Due to the odd/irregular shapes of the counties, some of the labels aren't in the perfect location. Ideally, you'd the labels to be in the 'visual center' of the polygons, or in an area inside the polygon where there is adequate room. You don't want any of the labels overlapping each other, and you want it to be visually obvious which label goes with which polygon.

In this example, Baltimore is at the south edge of its polygon, and slightly overlapping Howard. And Frederick is slightly overlapping Washington. Therefore let's move Baltimore up (north) slightly, and more Frederick down (south) slightly, by manipulating their Y-values. In addition to modifying the X/Y coordinates, you could also modify the 'position', the 'angle', and the size of the text when needed.

```
data anno_labels; set anno_labels;
if idname='Baltimore' then y=y+2;
if idname='Frederick' then y=y-2;
run;
```


Labels at Calculated Centroids
Garrett
Allegany
Washington
Frederick
Carroll
Baltimore
Harford
Cecil
Baltimore
Howard
Montgomery
Kent
Queen Anne's
Anne Arundel
Caroline
Prince George's
Talbot
Calvert
Charles
Dorchester
Wicomico
St. Mary's
Worcester
Somerset
Shifting Labels to Avoid Collisions

## Labels in the Ocean

Sometimes, try as you might with X/Y offsets and such, you still can't find a good place to put the labels on a map. For example, see the labels in the small northeast states in the US map below. When that happens, you might need to place the labels outside the map, and then draw a line to connect the label to the map area.




This example uses a specially-prepared dataset (maps.uscenter) provided by SAS. It contains X/Y centroids for all the US states, and also contains an alternate location to place certain labels in the ocean, for some of the small states in the northeast. This dataset is specifically designed to work with the maps.us map, and if you are using any other map then you will need to create your own dataset with the X/Y center & ocean locations for your map (not a trivial task).

Below are the first few observations of the maps.uscenter dataset. The rows where ocean='N' are the state centroids, and ocean='Y' are the alternate locations for the text label in the water. Note that the state values are the numeric fips codes, and the X/Y are projected values (not unprojected lat/long in degrees or radians), and they match up with the projected X/Y in maps.us.

**maps.uscenter**

| STATE | X | Y | OCEAN |
|---|---|---|---|
| 1 | 0.13521 | -0.06648 | N |
| 2 | -0.32189 | -0.12184 | N |
| 4 | -0.22291 | -0.02499 | N |
| 5 | 0.04777 | -0.03361 | N |
| 6 | -0.33100 | 0.04256 | N |
| 8 | -0.12985 | 0.04230 | N |
| 9 | 0.31165 | 0.10281 | Y |
| 9 | 0.29787 | 0.11741 | N |
| 10 | 0.29972 | 0.06737 | Y |
| 10 | 0.27260 | 0.06482 | N |

This special dataset, and the technique to use it has been around for ~20+ years, and the code below was slightly adapted from a very old Tech Support example. For the states that have an ocean location, the ocean X/Y is listed first in the dataset, and then the next observation is the land location (centroid of the state). Therefore if you encounter an observation where ocean='Y', you output the label at that (ocean) location, and then set a flag. Then as you process the next line of data, you find that the flag is set 'draw' a line to the X/Y land coordinate (rather than output a text label). The code is a little difficult to follow, but fortunately it is something that you can easily re-use without totally understanding it.

```
data anno_labels; set maps.uscenter;
length function $8 color $8;
retain flag 0;
xsys='2'; ysys='2'; hsys='3'; when='a';
function='label'; color='blue'; size=2.3;
if ocean^='Y' and flag^=1 then do;
 text=trim(left(fipstate(state))); position='5'; output;
 end;
else if ocean='Y' then do;
 text=trim(left(fipstate(state))); position='6'; output;
 function='move'; text=''; position=''; output;
 flag=1;
 end;
else if flag=1 then do;
```

```
 function='draw'; size=.25; output;
 flag=0;
 end;
run;
```

Here is what the resulting annotate dataset looks like (first 10 rows)…

**anno_labels**

| STATE | X | Y | OCEAN | flag | function | color | xsys | ysys | hsys | when | size | position | text |
|---|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | 0.13521 | -0.06648 | N | 0 | label | blue | 2 | 2 | 3 | a | 2.30 | 5 | AL |
| 2 | -0.32189 | -0.12184 | N | 0 | label | blue | 2 | 2 | 3 | a | 2.30 | 5 | AK |
| 4 | -0.22291 | -0.02499 | N | 0 | label | blue | 2 | 2 | 3 | a | 2.30 | 5 | AZ |
| 5 | 0.04777 | -0.03361 | N | 0 | label | blue | 2 | 2 | 3 | a | 2.30 | 5 | AR |
| 6 | -0.33100 | 0.04256 | N | 0 | label | blue | 2 | 2 | 3 | a | 2.30 | 5 | CA |
| 8 | -0.12985 | 0.04230 | N | 0 | label | blue | 2 | 2 | 3 | a | 2.30 | 5 | CO |
| 9 | 0.31165 | 0.10281 | Y | 0 | label | blue | 2 | 2 | 3 | a | 2.30 | 6 | CT |
| 9 | 0.31165 | 0.10281 | Y | 0 | move | blue | 2 | 2 | 3 | a | 2.30 | | |
| 9 | 0.29787 | 0.11741 | N | 1 | draw | blue | 2 | 2 | 3 | a | 0.25 | | |
| 10 | 0.29972 | 0.06737 | Y | 0 | label | blue | 2 | 2 | 3 | a | 2.30 | 6 | DE |
| 10 | 0.29972 | 0.06737 | Y | 0 | move | blue | 2 | 2 | 3 | a | 2.30 | | |
| 10 | 0.27260 | 0.06482 | N | 1 | draw | blue | 2 | 2 | 3 | a | 0.25 | | |

The map looks much better now – notice that none of the labels overlap, and it is much easier to read than the previous version.




On the next few pages you will find a few fancier real-world examples that use these techniques to annotate labels on maps. Hopefully these will give you a few ideas as to how you can put these techniques to use!


States Visited by Johnboy Sasman
WA
OR
CA
ID
NV
MT
WY
UT
AZ
CO
NM
ND
SD
NE
KS
OK
TX
MN
IA
MO
AR
LA
WI
IL
MS
MI
IN
KY
TN
AL
OH
WV
GA
FL
SC
NC
VA
PA
NY
ME
VT
NH
MA
RI
CT
NJ
DE
MD
AK
HI
PR


Throughout the month of May, the Census Bureau released new information about income, education, commuting and birthplace for every state.
Click on a state to see the latest information.
The Census Bureau also released a demographic profile for the USA.
U.S. demographic profile
U.S. population center
Washington
Oregon
California
Idaho
Nevada
Montana
Wyoming
Utah
Arizona
Colorado
New Mexico
North Dakota
South Dakota
Nebraska
Kansas
Oklahoma
Texas
Minnesota
Iowa
Missouri
Arkansas
Louisiana
Wisconsin
Illinois
Michigan
Indiana
Ohio
Kentucky
Tennessee
Mississippi
Alabama
Georgia
Florida
South Carolina
North Carolina
Virginia
West Virginia
Pennsylvania
New York
Maine
Vermont
New Hampshire
Massachusetts
Rhode Island
Connecticut
New Jersey
Delaware
Maryland
Alaska
Hawaii
SAS/Graph Imitation of USA Today map

**August 19, 2013**
**Persons infected with the outbreak strain of *Salmonella Typhimurium*, by State**




**Figure 1. Percent Change in Resident Population for the 50 States, the District of Columbia, and Puerto Rico: 1990 to 2000**



## North Carolina Emergency Management Areas

State of NC and ARES




Hover over county for name, branch, and division.
Click county to go to ARES County EC.
Click bullseye for map to Branch Office

## 16 States of Germany


Changes in Life Expectancy
Between 1970 and 2010
+7.5 years
-2.3
+14.3
+7.9
+1.1
+13.2
+12.9
+3.5
+9.0
Belarus
-1.2
Increase in alcohol consumption
Iran
+21.3
Improvements in health programs
Bangladesh
+24.1
Improvements in health programs
Maldives
+28.4
Improvements in health programs
Lesotho
-5.5
Increase in HIV
Average Global Life Expectancy
70.4
Female
73.3
Male
67.5